// File: contracts/OpolisPay.sol



pragma solidity 0.8.5;

// SPDX-License-Identifier: LGPLv3
import "@openzeppelin/contracts/token/ERC20/IERC20.sol";
import "@openzeppelin/contracts/token/ERC20/utils/SafeERC20.sol";
import "@openzeppelin/contracts/security/ReentrancyGuard.sol";

/// @notice custom errors for revert statements

/// @dev requires privileged access
error NotPermitted();

/// @dev not a whitelisted token
error NotWhitelisted();

/// @dev payroll id equals zero
error InvalidPayroll();

/// @dev payroll id used
error DuplicatePayroll();

/// @dev amount equals zero
error InvalidAmount();

/// @dev sender is not a member
error NotMember();

/// @dev stake must be a non zero amount of whitelisted token
/// or non zero amount of eth
error InvalidStake();

/// @dev stake must be a non zero amount of whitelisted token
/// or non zero amount of eth
error InvalidWithdraw();

/// @dev setting one of the role to zero address
error ZeroAddress();

/// @dev withdrawing non whitelisted token
error InvalidToken();

/// @dev whitelisting and empty list of tokens
error ZeroTokens();

/// @dev token has already been whitelisted
error AlreadyWhitelisted();

/// @dev sending eth directly to contract address
error DirectTransfer();

/// @dev token and destination length mismatch
error LengthMismatch();

/// @title OpolisPay
/// @notice Minimalist Contract for Crypto Payroll Payments
contract OpolisPay is ReentrancyGuard {
    using SafeERC20 for IERC20;

    address internal constant ETH = 0xEeeeeEeeeEeEeeEeEeEeeEEEeeeeEeeeeeeeEEeE;
    address internal constant ZERO = address(0);

    address[] public supportedTokens; //Tokens that can be sent.
    address private opolisAdmin; //Should be Opolis multi-sig for security
    address private opolisHelper; //Can be bot wallet for convenience
    address private ethLiquidation; //Address for ETH liquidations

    event SetupComplete(
        address indexed admin,
        address indexed helper,
        address ethLiquidation,
        address[] tokens,
        address[] liqDestinations
    );
    event Staked(
        address indexed staker, address indexed token, uint256 amount, uint256 indexed memberId, uint256 stakeNumber
    );
    event Paid(address indexed payor, address indexed token, uint256 indexed payrollId, uint256 amount);
    event OpsPayrollWithdraw(address indexed token, uint256 indexed payrollId, uint256 amount);
    event OpsStakeWithdraw(address indexed token, uint256 indexed stakeId, uint256 stakeNumber, uint256 amount);
    event Sweep(address indexed token, uint256 amount);
    event NewDestination(address indexed oldDestination, address indexed token, address indexed destination);
    event NewAdmin(address indexed oldAdmin, address indexed opolisAdmin);
    event NewHelper(address indexed oldHelper, address indexed newHelper);
    event NewTokens(address[] newTokens, address[] newDestinations);

    mapping(uint256 => uint256) private stakes; //Tracks used stake ids
    mapping(uint256 => bool) private payrollIds; //Tracks used payroll ids
    mapping(uint256 => bool) public payrollWithdrawn; //Tracks payroll withdrawals
    mapping(uint256 => uint256) public stakeWithdrawn; //Tracks stake withdrawals
    mapping(address => bool) public whitelisted; //Tracks whitelisted tokens
    mapping(address => address) public liqDestinations; //Tracks liquidation destinations for tokens

    modifier onlyAdmin() {
        if (msg.sender != opolisAdmin) revert NotPermitted();
        _;
    }

    modifier onlyOpolis() {
        if (!(msg.sender == opolisAdmin || msg.sender == opolisHelper)) {
            revert NotPermitted();
        }
        _;
    }

    /// @notice launches contract with a destination as the Opolis wallet, the admins, and a token whitelist
    /// @param _opolisAdmin the multi-sig which is the ultimate admin
    /// @param _opolisHelper meant to allow for a bot to handle less sensitive items
    /// @param _ethLiq the address where we send eth or native token liquidations
    /// @param _tokenList initial whitelist of tokens for staking and payroll
    /// @param _destinationList the addresses where payroll and stakes will be sent when withdrawn based on token

    constructor(
        address _opolisAdmin,
        address _opolisHelper,
        address _ethLiq,
        address[] memory _tokenList,
        address[] memory _destinationList
    ) {
        if (_tokenList.length != _destinationList.length) revert LengthMismatch();
        opolisAdmin = _opolisAdmin;
        opolisHelper = _opolisHelper;
        ethLiquidation = _ethLiq;

        for (uint256 i = 0; i < _tokenList.length; i++) {
            _addToken(_tokenList[i]);
            _addDestination(_destinationList[i], _tokenList[i]);
        }

        emit SetupComplete(opolisAdmin, opolisHelper, _ethLiq, _tokenList, _destinationList);
    }

    /**
     *
     *                          CORE PAYROLL FUNCTIONS
     *
     */

    /// @notice core function for members to pay their payroll
    /// @param token the token being used to pay for their payroll
    /// @param amount the amount due for their payroll -- up to user / front-end to match
    /// @param payrollId the way we'll associate payments with members' invoices

    function payPayroll(address token, uint256 amount, uint256 payrollId) external nonReentrant {
        if (!whitelisted[token]) revert NotWhitelisted();
        if (payrollId == 0) revert InvalidPayroll();
        if (amount == 0) revert InvalidAmount();
        if (payrollIds[payrollId]) revert DuplicatePayroll();

        IERC20(token).safeTransferFrom(msg.sender, address(this), amount);
        payrollIds[payrollId] = true;

        emit Paid(msg.sender, token, payrollId, amount);
    }

    /// @notice staking function that allows for both ETH and whitelisted ERC20
    /// @param token the token being used to stake
    /// @param amount the amount due for staking -- up to user / front-end to match
    /// @param memberId the way we'll associate the stake with a new member

    function memberStake(address token, uint256 amount, uint256 memberId) public payable nonReentrant {
        if (!((whitelisted[token] && amount != 0) || (token == ETH && msg.value != 0))) {
            revert InvalidStake();
        }
        if (memberId == 0) revert NotMember();

        // @dev increments the stake id for each member
        uint256 stakeCount = ++stakes[memberId];

        // @dev function for auto transfering out stake
        if (msg.value > 0 && token == ETH) {
            payable(ethLiquidation).transfer(msg.value);
            emit Staked(msg.sender, ETH, msg.value, memberId, stakeCount);
        } else {
            IERC20(token).safeTransferFrom(msg.sender, address(this), amount);
            emit Staked(msg.sender, token, amount, memberId, stakeCount);
        }
    }

    /// @notice withdraw function for admin or OpsBot to call
    /// @param _payrollIds the paid payrolls we want to clear out
    /// @param _payrollTokens the tokens the payrolls were paid in
    /// @param _payrollAmounts the amount that was paid
    /// @dev we iterate through payrolls and clear them out with the funds being sent to the destination address
    function withdrawPayrolls(
        uint256[] calldata _payrollIds,
        address[] calldata _payrollTokens,
        uint256[] calldata _payrollAmounts
    ) external onlyOpolis {
        uint256[] memory withdrawAmounts = new uint256[](
            supportedTokens.length
        );
        for (uint256 i = 0; i < _payrollIds.length; i++) {
            uint256 id = _payrollIds[i];
            if (!payrollIds[id]) revert InvalidPayroll();

            address token = _payrollTokens[i];
            uint256 amount = _payrollAmounts[i];

            if (!payrollWithdrawn[id]) {
                uint256 j;
                for (; j < supportedTokens.length; j++) {
                    if (supportedTokens[j] == token) {
                        withdrawAmounts[j] += amount;
                        break;
                    }
                }
                if (j == supportedTokens.length) revert InvalidToken();
                payrollWithdrawn[id] = true;

                emit OpsPayrollWithdraw(token, id, amount);
            }
        }

        for (uint256 i = 0; i < withdrawAmounts.length; i++) {
            uint256 amount = withdrawAmounts[i];
            if (amount > 0) {
                _withdraw(supportedTokens[i], amount);
            }
        }
    }

    /// @notice withdraw function for admin or OpsBot to call
    /// @param _stakeIds the paid stakes id we want to clear out
    /// @param _stakeNum the particular stake number associated with that id
    /// @param _stakeTokens the tokens the stakes were paid in
    /// @param _stakeAmounts the amount that was paid
    /// @dev we iterate through stakes and clear them out with the funds being sent to the destination address
    function withdrawStakes(
        uint256[] calldata _stakeIds,
        uint256[] calldata _stakeNum,
        address[] calldata _stakeTokens,
        uint256[] calldata _stakeAmounts
    ) external onlyOpolis {
        uint256[] memory withdrawAmounts = new uint256[](
            supportedTokens.length
        );
        if (_stakeIds.length != _stakeNum.length) revert InvalidWithdraw();

        for (uint256 i = 0; i < _stakeIds.length; i++) {
            uint256 id = _stakeIds[i];
            address token = _stakeTokens[i];
            uint256 amount = _stakeAmounts[i];
            uint256 num = _stakeNum[i];

            if (stakeWithdrawn[id] < num) {
                uint256 j;
                for (; j < supportedTokens.length; j++) {
                    if (supportedTokens[j] == token) {
                        withdrawAmounts[j] += amount;
                        break;
                    }
                }
                if (j == supportedTokens.length) revert InvalidToken();
                stakeWithdrawn[id] = num;

                emit OpsStakeWithdraw(token, id, num, amount);
            }
        }

        for (uint256 i = 0; i < withdrawAmounts.length; i++) {
            uint256 amount = withdrawAmounts[i];
            if (amount > 0) {
                _withdraw(supportedTokens[i], amount);
            }
        }
    }

    /// @notice clearBalance() is meant to be a safety function to be used for stuck funds or upgrades
    /// @dev will mark any non-withdrawn payrolls as withdrawn

    function clearBalance() public onlyAdmin {
        for (uint256 i = 0; i < supportedTokens.length; i++) {
            address token = supportedTokens[i];
            uint256 balance = IERC20(token).balanceOf(address(this));

            if (balance > 0) {
                _withdraw(token, balance);
            }
            emit Sweep(token, balance);
        }
    }

    /// @notice fallback function to prevent accidental ether transfers
    /// @dev if someone tries to send ether directly to the contract the tx will fail

    receive() external payable {
        revert DirectTransfer();
    }

    /**
     *
     *                          ADMIN FUNCTIONS
     *
     */

    /// @notice this function is used to adjust where member funds are sent by contract
    /// @param token since each token has a new destination
    /// @param newDestination is the new address where funds are sent (assumes it's payable exchange address)
    /// @dev must be called by Opolis Admin multi-sig

    function updateDestination(address token, address newDestination) external onlyAdmin {
        if (newDestination == ZERO) revert ZeroAddress();

        address oldDestination = liqDestinations[token];
        liqDestinations[token] = newDestination;

        emit NewDestination(oldDestination, token, newDestination);
    }

    /// @notice this function is used to replace the admin multi-sig
    /// @param newAdmin is the new admin address
    /// @dev this should always be a mulit-sig

    function updateAdmin(address newAdmin) external onlyAdmin returns (address) {
        if (newAdmin == ZERO) revert ZeroAddress();

        emit NewAdmin(opolisAdmin, newAdmin);
        opolisAdmin = newAdmin;

        return opolisAdmin;
    }

    /// @notice this function is used to replace a bot
    /// @param newHelper is the new bot address
    /// @dev this can be a hot wallet, since it has limited powers

    function updateHelper(address newHelper) external onlyAdmin returns (address) {
        if (newHelper == ZERO) revert ZeroAddress();

        emit NewHelper(opolisHelper, newHelper);
        opolisHelper = newHelper;

        return opolisHelper;
    }

    /// @notice this function is used to add new whitelisted tokens
    /// @param newTokens are the tokens to be whitelisted
    /// @param newDestinations since each new token may have a different destination
    /// @dev restricted to admin b/c this is a business / compliance decision

    function addTokens(address[] memory newTokens, address[] memory newDestinations) external onlyAdmin {
        if (newTokens.length == 0) revert ZeroTokens();
        if (newTokens.length != newDestinations.length) revert LengthMismatch();

        for (uint256 i = 0; i < newTokens.length; i++) {
            _addToken(newTokens[i]);
            _addDestination(newDestinations[i], newTokens[i]);
        }

        emit NewTokens(newTokens, newDestinations);
    }

    /**
     *
     *                          INTERNAL FUNCTIONS
     *
     */

    function _addToken(address token) internal {
        if (whitelisted[token]) revert AlreadyWhitelisted();
        if (token == ZERO) revert ZeroAddress();
        supportedTokens.push(token);
        whitelisted[token] = true;
    }

    function _addDestination(address destination, address token) internal {
        if (destination == ZERO) revert ZeroAddress();
        liqDestinations[token] = destination;
    }

    function _withdraw(address token, uint256 amount) internal {
        address dest = liqDestinations[token];
        IERC20(token).safeTransfer(dest, amount);
    }
}

// File: @openzeppelin/contracts/token/ERC20/IERC20.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (token/ERC20/IERC20.sol)

pragma solidity ^0.8.0;

/**
 * @dev Interface of the ERC20 standard as defined in the EIP.
 */
interface IERC20 {
    /**
     * @dev Returns the amount of tokens in existence.
     */
    function totalSupply() external view returns (uint256);

    /**
     * @dev Returns the amount of tokens owned by `account`.
     */
    function balanceOf(address account) external view returns (uint256);

    /**
     * @dev Moves `amount` tokens from the caller's account to `recipient`.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * Emits a {Transfer} event.
     */
    function transfer(address recipient, uint256 amount) external returns (bool);

    /**
     * @dev Returns the remaining number of tokens that `spender` will be
     * allowed to spend on behalf of `owner` through {transferFrom}. This is
     * zero by default.
     *
     * This value changes when {approve} or {transferFrom} are called.
     */
    function allowance(address owner, address spender) external view returns (uint256);

    /**
     * @dev Sets `amount` as the allowance of `spender` over the caller's tokens.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * IMPORTANT: Beware that changing an allowance with this method brings the risk
     * that someone may use both the old and the new allowance by unfortunate
     * transaction ordering. One possible solution to mitigate this race
     * condition is to first reduce the spender's allowance to 0 and set the
     * desired value afterwards:
     * https://github.com/ethereum/EIPs/issues/20#issuecomment-263524729
     *
     * Emits an {Approval} event.
     */
    function approve(address spender, uint256 amount) external returns (bool);

    /**
     * @dev Moves `amount` tokens from `sender` to `recipient` using the
     * allowance mechanism. `amount` is then deducted from the caller's
     * allowance.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * Emits a {Transfer} event.
     */
    function transferFrom(
        address sender,
        address recipient,
        uint256 amount
    ) external returns (bool);

    /**
     * @dev Emitted when `value` tokens are moved from one account (`from`) to
     * another (`to`).
     *
     * Note that `value` may be zero.
     */
    event Transfer(address indexed from, address indexed to, uint256 value);

    /**
     * @dev Emitted when the allowance of a `spender` for an `owner` is set by
     * a call to {approve}. `value` is the new allowance.
     */
    event Approval(address indexed owner, address indexed spender, uint256 value);
}

// File: @openzeppelin/contracts/token/ERC20/utils/SafeERC20.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (token/ERC20/utils/SafeERC20.sol)

pragma solidity ^0.8.0;

import "../IERC20.sol";
import "../../../utils/Address.sol";

/**
 * @title SafeERC20
 * @dev Wrappers around ERC20 operations that throw on failure (when the token
 * contract returns false). Tokens that return no value (and instead revert or
 * throw on failure) are also supported, non-reverting calls are assumed to be
 * successful.
 * To use this library you can add a `using SafeERC20 for IERC20;` statement to your contract,
 * which allows you to call the safe operations as `token.safeTransfer(...)`, etc.
 */
library SafeERC20 {
    using Address for address;

    function safeTransfer(
        IERC20 token,
        address to,
        uint256 value
    ) internal {
        _callOptionalReturn(token, abi.encodeWithSelector(token.transfer.selector, to, value));
    }

    function safeTransferFrom(
        IERC20 token,
        address from,
        address to,
        uint256 value
    ) internal {
        _callOptionalReturn(token, abi.encodeWithSelector(token.transferFrom.selector, from, to, value));
    }

    /**
     * @dev Deprecated. This function has issues similar to the ones found in
     * {IERC20-approve}, and its usage is discouraged.
     *
     * Whenever possible, use {safeIncreaseAllowance} and
     * {safeDecreaseAllowance} instead.
     */
    function safeApprove(
        IERC20 token,
        address spender,
        uint256 value
    ) internal {
        // safeApprove should only be called when setting an initial allowance,
        // or when resetting it to zero. To increase and decrease it, use
        // 'safeIncreaseAllowance' and 'safeDecreaseAllowance'
        require(
            (value == 0) || (token.allowance(address(this), spender) == 0),
            "SafeERC20: approve from non-zero to non-zero allowance"
        );
        _callOptionalReturn(token, abi.encodeWithSelector(token.approve.selector, spender, value));
    }

    function safeIncreaseAllowance(
        IERC20 token,
        address spender,
        uint256 value
    ) internal {
        uint256 newAllowance = token.allowance(address(this), spender) + value;
        _callOptionalReturn(token, abi.encodeWithSelector(token.approve.selector, spender, newAllowance));
    }

    function safeDecreaseAllowance(
        IERC20 token,
        address spender,
        uint256 value
    ) internal {
        unchecked {
            uint256 oldAllowance = token.allowance(address(this), spender);
            require(oldAllowance >= value, "SafeERC20: decreased allowance below zero");
            uint256 newAllowance = oldAllowance - value;
            _callOptionalReturn(token, abi.encodeWithSelector(token.approve.selector, spender, newAllowance));
        }
    }

    /**
     * @dev Imitates a Solidity high-level call (i.e. a regular function call to a contract), relaxing the requirement
     * on the return value: the return value is optional (but if data is returned, it must not be false).
     * @param token The token targeted by the call.
     * @param data The call data (encoded using abi.encode or one of its variants).
     */
    function _callOptionalReturn(IERC20 token, bytes memory data) private {
        // We need to perform a low level call here, to bypass Solidity's return data size checking mechanism, since
        // we're implementing it ourselves. We use {Address.functionCall} to perform this call, which verifies that
        // the target address contains contract code and also asserts for success in the low-level call.

        bytes memory returndata = address(token).functionCall(data, "SafeERC20: low-level call failed");
        if (returndata.length > 0) {
            // Return data is optional
            require(abi.decode(returndata, (bool)), "SafeERC20: ERC20 operation did not succeed");
        }
    }
}

// File: @openzeppelin/contracts/security/ReentrancyGuard.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (security/ReentrancyGuard.sol)

pragma solidity ^0.8.0;

/**
 * @dev Contract module that helps prevent reentrant calls to a function.
 *
 * Inheriting from `ReentrancyGuard` will make the {nonReentrant} modifier
 * available, which can be applied to functions to make sure there are no nested
 * (reentrant) calls to them.
 *
 * Note that because there is a single `nonReentrant` guard, functions marked as
 * `nonReentrant` may not call one another. This can be worked around by making
 * those functions `private`, and then adding `external` `nonReentrant` entry
 * points to them.
 *
 * TIP: If you would like to learn more about reentrancy and alternative ways
 * to protect against it, check out our blog post
 * https://blog.openzeppelin.com/reentrancy-after-istanbul/[Reentrancy After Istanbul].
 */
abstract contract ReentrancyGuard {
    // Booleans are more expensive than uint256 or any type that takes up a full
    // word because each write operation emits an extra SLOAD to first read the
    // slot's contents, replace the bits taken up by the boolean, and then write
    // back. This is the compiler's defense against contract upgrades and
    // pointer aliasing, and it cannot be disabled.

    // The values being non-zero value makes deployment a bit more expensive,
    // but in exchange the refund on every call to nonReentrant will be lower in
    // amount. Since refunds are capped to a percentage of the total
    // transaction's gas, it is best to keep them low in cases like this one, to
    // increase the likelihood of the full refund coming into effect.
    uint256 private constant _NOT_ENTERED = 1;
    uint256 private constant _ENTERED = 2;

    uint256 private _status;

    constructor() {
        _status = _NOT_ENTERED;
    }

    /**
     * @dev Prevents a contract from calling itself, directly or indirectly.
     * Calling a `nonReentrant` function from another `nonReentrant`
     * function is not supported. It is possible to prevent this from happening
     * by making the `nonReentrant` function external, and making it call a
     * `private` function that does the actual work.
     */
    modifier nonReentrant() {
        // On the first call to nonReentrant, _notEntered will be true
        require(_status != _ENTERED, "ReentrancyGuard: reentrant call");

        // Any calls to nonReentrant after this point will fail
        _status = _ENTERED;

        _;

        // By storing the original value once again, a refund is triggered (see
        // https://eips.ethereum.org/EIPS/eip-2200)
        _status = _NOT_ENTERED;
    }
}

// File: @openzeppelin/contracts/utils/Address.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (utils/Address.sol)

pragma solidity ^0.8.0;

/**
 * @dev Collection of functions related to the address type
 */
library Address {
    /**
     * @dev Returns true if `account` is a contract.
     *
     * [IMPORTANT]
     * ====
     * It is unsafe to assume that an address for which this function returns
     * false is an externally-owned account (EOA) and not a contract.
     *
     * Among others, `isContract` will return false for the following
     * types of addresses:
     *
     *  - an externally-owned account
     *  - a contract in construction
     *  - an address where a contract will be created
     *  - an address where a contract lived, but was destroyed
     * ====
     */
    function isContract(address account) internal view returns (bool) {
        // This method relies on extcodesize, which returns 0 for contracts in
        // construction, since the code is only stored at the end of the
        // constructor execution.

        uint256 size;
        assembly {
            size := extcodesize(account)
        }
        return size > 0;
    }

    /**
     * @dev Replacement for Solidity's `transfer`: sends `amount` wei to
     * `recipient`, forwarding all available gas and reverting on errors.
     *
     * https://eips.ethereum.org/EIPS/eip-1884[EIP1884] increases the gas cost
     * of certain opcodes, possibly making contracts go over the 2300 gas limit
     * imposed by `transfer`, making them unable to receive funds via
     * `transfer`. {sendValue} removes this limitation.
     *
     * https://diligence.consensys.net/posts/2019/09/stop-using-soliditys-transfer-now/[Learn more].
     *
     * IMPORTANT: because control is transferred to `recipient`, care must be
     * taken to not create reentrancy vulnerabilities. Consider using
     * {ReentrancyGuard} or the
     * https://solidity.readthedocs.io/en/v0.5.11/security-considerations.html#use-the-checks-effects-interactions-pattern[checks-effects-interactions pattern].
     */
    function sendValue(address payable recipient, uint256 amount) internal {
        require(address(this).balance >= amount, "Address: insufficient balance");

        (bool success, ) = recipient.call{value: amount}("");
        require(success, "Address: unable to send value, recipient may have reverted");
    }

    /**
     * @dev Performs a Solidity function call using a low level `call`. A
     * plain `call` is an unsafe replacement for a function call: use this
     * function instead.
     *
     * If `target` reverts with a revert reason, it is bubbled up by this
     * function (like regular Solidity function calls).
     *
     * Returns the raw returned data. To convert to the expected return value,
     * use https://solidity.readthedocs.io/en/latest/units-and-global-variables.html?highlight=abi.decode#abi-encoding-and-decoding-functions[`abi.decode`].
     *
     * Requirements:
     *
     * - `target` must be a contract.
     * - calling `target` with `data` must not revert.
     *
     * _Available since v3.1._
     */
    function functionCall(address target, bytes memory data) internal returns (bytes memory) {
        return functionCall(target, data, "Address: low-level call failed");
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`], but with
     * `errorMessage` as a fallback revert reason when `target` reverts.
     *
     * _Available since v3.1._
     */
    function functionCall(
        address target,
        bytes memory data,
        string memory errorMessage
    ) internal returns (bytes memory) {
        return functionCallWithValue(target, data, 0, errorMessage);
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
     * but also transferring `value` wei to `target`.
     *
     * Requirements:
     *
     * - the calling contract must have an ETH balance of at least `value`.
     * - the called Solidity function must be `payable`.
     *
     * _Available since v3.1._
     */
    function functionCallWithValue(
        address target,
        bytes memory data,
        uint256 value
    ) internal returns (bytes memory) {
        return functionCallWithValue(target, data, value, "Address: low-level call with value failed");
    }

    /**
     * @dev Same as {xref-Address-functionCallWithValue-address-bytes-uint256-}[`functionCallWithValue`], but
     * with `errorMessage` as a fallback revert reason when `target` reverts.
     *
     * _Available since v3.1._
     */
    function functionCallWithValue(
        address target,
        bytes memory data,
        uint256 value,
        string memory errorMessage
    ) internal returns (bytes memory) {
        require(address(this).balance >= value, "Address: insufficient balance for call");
        require(isContract(target), "Address: call to non-contract");

        (bool success, bytes memory returndata) = target.call{value: value}(data);
        return verifyCallResult(success, returndata, errorMessage);
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
     * but performing a static call.
     *
     * _Available since v3.3._
     */
    function functionStaticCall(address target, bytes memory data) internal view returns (bytes memory) {
        return functionStaticCall(target, data, "Address: low-level static call failed");
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-string-}[`functionCall`],
     * but performing a static call.
     *
     * _Available since v3.3._
     */
    function functionStaticCall(
        address target,
        bytes memory data,
        string memory errorMessage
    ) internal view returns (bytes memory) {
        require(isContract(target), "Address: static call to non-contract");

        (bool success, bytes memory returndata) = target.staticcall(data);
        return verifyCallResult(success, returndata, errorMessage);
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
     * but performing a delegate call.
     *
     * _Available since v3.4._
     */
    function functionDelegateCall(address target, bytes memory data) internal returns (bytes memory) {
        return functionDelegateCall(target, data, "Address: low-level delegate call failed");
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-string-}[`functionCall`],
     * but performing a delegate call.
     *
     * _Available since v3.4._
     */
    function functionDelegateCall(
        address target,
        bytes memory data,
        string memory errorMessage
    ) internal returns (bytes memory) {
        require(isContract(target), "Address: delegate call to non-contract");

        (bool success, bytes memory returndata) = target.delegatecall(data);
        return verifyCallResult(success, returndata, errorMessage);
    }

    /**
     * @dev Tool to verifies that a low level call was successful, and revert if it wasn't, either by bubbling the
     * revert reason using the provided one.
     *
     * _Available since v4.3._
     */
    function verifyCallResult(
        bool success,
        bytes memory returndata,
        string memory errorMessage
    ) internal pure returns (bytes memory) {
        if (success) {
            return returndata;
        } else {
            // Look for revert reason and bubble it up if present
            if (returndata.length > 0) {
                // The easiest way to bubble the revert reason is using memory via assembly

                assembly {
                    let returndata_size := mload(returndata)
                    revert(add(32, returndata), returndata_size)
                }
            } else {
                revert(errorMessage);
            }
        }
    }
}